The Mathematica® Journal

Some Integrals Involving Symmetric-Top Eigenfunctions

Russell A. Bonham

Using closure over a complete set of rotational states for methane to evaluate the intensity for quasi-elastic electron scattering in the first Born approximation, a number of integrals were encountered that appear not to have been evaluated previously. *Mathematica* was employed to evaluate these and similar integrals, and it was discovered that in all the cases studied the results could be represented by simple formulas.

■ Details of the Integral Evaluations

The integrals discussed here are related to averages over the spherical matrix elements $D^J_{M_1 M_2}(a, x, \gamma)$ [1, 2] defined by

$$D^J_{M_1 M_2}(\alpha, x, \gamma) = e^{-i\alpha M_1}\, d^J_{M_1 M_2}(x)\, e^{-i\gamma M_2}, \tag{1}$$

which are symmetric-top eigenfunctions with eigenvalues $J(J+1)$ and a degeneracy $(2J+1)^2$. The function $d^J_{M_1 M_2}(x)$ is defined [1, 2] as

$$d^J_{M_1 M_2}(x) = \sum_t (-1)^t \frac{\sqrt{(J+M_1)!\,(J-M_1)!\,(J+M_2)!\,(J-M_2)!}}{2^J\,(J+M_1-t)!\,(J-M_2-t)!\,t!\,(t+M_2-M_1)!} \tag{2}$$
$$(1+x)^{J+\frac{M_1-M_2}{2}-t}\,(1-x)^{t-\frac{M_1-M_2}{2}},$$

where the sum is over all integer values of t from 0 to the first negative factorial that occurs in the denominator. In the impulse approximation for quasi-elastic electron scattering [3], averages over the rotational motion of a spherical top molecule such as methane are integrals of the following form:

$$I_{J M_1 M_2}(n) = \frac{1}{2}\int_{-1}^{1} d^J_{M_1 M_2}(x)\left(1-x^2\right)^n H(x)\, d^J_{M_1 M_2}(x)\, dx, \tag{3}$$

where

$$H = -\left(1 - x^2\right) \frac{d^2}{dx^2} - 2\,x \frac{d}{dx} \tag{4}$$

is the Hamiltonian operator whose eigenfunctions are the Legendre polynomials $P_J(x)$ with eigenvalues $J(J+1)$. Equation (4) is also a part of the Hamiltonian for which the $D^J_{M_1 M_2}(a, x, \gamma)$ are solutions, but the $d^J_{M_1 M_2}(x)$ are not eigenfunctions of $H(x)$. The occurrence of H arises in the treatment of the scattering problem in the impulse approximation [3]. By use of *Mathematica* it was possible to prove that

$$\sum_{M_1=-J}^{J} \sum_{M_2=-J}^{J} N_{J M_1 M_2}\, I_{J M_1 M_2}(0) = \frac{1}{3} J(J+1)\,(2\,J+1)^2, \tag{5}$$

where the normalization factor $N_{J M_1 M_2}$ is given by

$$N_{J M_1 M_2} = \frac{1}{\frac{1}{2}\int_{-1}^{1} d^J_{M_1 M_2}(x)\, d^J_{M_1 M_2}(x)\, dx} = 2\,J + 1 \tag{6}$$

for all values of M_1 and M_2. For integer $n > 0$ the results are

$$\sum_{M_1=-J}^{J} \sum_{M_2=-J}^{J} N_{J M_1 M_2}\, I_{J M_1 M_2}(n) = C(n)\, J\,(J+1)\,(2\,J+1)^2, \tag{7}$$

where

$$C(n) = \frac{1}{3} \frac{2^n\, n!}{(2\,n+1)!!}. \tag{8}$$

The use of the word "prove" is based on the results from a *Mathematica* program. All of these expressions occur as ratios of positive integers and were found to be in exact agreement with the results on the right-hand sides in equation (7) for values of J from 0 through 18.

Note that the same result can be obtained for the case where $\left(1 - x^2\right)^n$ is replaced by $x^{2\,n}$ if $C(n)$ is replaced by

$$C(n) = \frac{1}{3\,(2\,n+1)}. \tag{9}$$

Except for the factor of $\frac{1}{3}$ for $\left(1 - x^2\right)^n$ and for $x^{2\,n}$ in the definitions of the $C(n)$, the remaining parts are equivalent to the integrals

$$\frac{1}{2} \int_{-1}^{1} \left(1 - x^2\right)^n dx = \frac{2^n\, n!}{(2\,n+1)!!} \tag{10}$$

and

$$\frac{1}{2} \int_{-1}^{1} x^{2\,n}\, dx = \frac{1}{(2\,n+1)}. \tag{11}$$

Further investigation using $\left(\sqrt{1-x^2}\right)^{2n+1}$ found the same result with

$$C(n) = \frac{1}{3}\frac{\pi(2n+1)!!}{2^{n+2}(n+1)!}, \tag{12}$$

where it is noted that

$$\frac{1}{2}\int_{-1}^{1}\left(\sqrt{1-x^2}\right)^{(2n+1)} dx = \frac{\pi(2n+1)!!}{2^{n+2}(n+1)!}. \tag{13}$$

Define the double sum over the integral

$$I_J(m,n) = \sum_{M_1=-J}^{J}\sum_{M_2=-J}^{J}\frac{1}{2}\int_{-1}^{1} d^J_{M_1 M_2}(x)\left(1-x^2\right)x^{2n}\,H d^J_{M_1 M_2}(x)\,dx, \tag{14}$$

and

$$I_C(m,n) = J(J+1)(2J+1)^2\,\frac{1}{2}\int_{-1}^{1}\left(1-x^2\right)x^{2n}\,dx. \tag{15}$$

These results suggest the conjecture that $I_J(m,n) = C_0\, I_C(m,n)$, where C_0 is a constant. For $m = 0, 1$ and $n = 0, 1$, the constant factor C_0 was found to be $\frac{1}{3}$ for $J = 1, 2, 3, 4$, as shown in the next section.

Implementation

The function `d` is $d^J_{M_1 M_2}(x)$.

```
d[j_, m1_, m2_, x_] := Module[{t, tmin, tmax},
  tmin = Max[m1 - m2, 0];
  tmax = Max[Min[j + m1, j - m2], tmin];
  Sqrt[(j + m1) ! (j - m1) ! (j + m2) ! (j - m2) !] / 2^j
   Sum[(-1)^t (1 + x)^(j - t + (m1 - m2) / 2)
     (1 - x)^(t - (m1 - m2) / 2) /
      ((j + m1 - t) ! (j - m2 - t) ! t ! (t - 2 (m1 - m2) / 2) !),
    {t, tmin, tmax}]]
```

The functions `d1` and `d2` are the first and second derivatives of $d^J_{M_1 M_2}(x)$ with respect to x.

```
d1[j_, m1_, m2_, x_] := D[d[j, m1, m2, x], x]
d2[j_, m1_, m2_, x_] := D[d1[j, m1, m2, x], x]
```

The function h is the result of the Hamiltonian H operating on d.

```
h[j_, m1_, m2_, x_] :=
 (1 - x^2) d2[j, m1, m2, x] - 2 x d1[j, m1, m2, x]
```

Define the integral a.

```
a[m_, n_] :=
 Module[{x},
  1 / 2 Integrate[(1 - x^2) ^ (m / 2) (x^2) ^n, {x, -1, 1}]]
```

The function int1 is the integral in equation (14).

```
int1[j_, m1_, m2_, m_, n_] :=
 Module[{x},
  1 / 2 Integrate[(1 - x^2) ^ (m / 2) (x^2) ^n d[j, m1, m2, x]
     h[j, m1, m2, x], {x, -1, 1}]]
```

The function int2 is the normalization integral of the square of $d^J_{M_1 M_2}(x)$.

```
int2[j_, m1_, m2_] :=
 Module[{x}, 1 / 2 Integrate[d[j, m1, m2, x]^2, {x, -1, 1}]]
```

The function s is the sum of the ratios of the integrals int1 and int2 over M_1 and M_2.

```
s[j_, m_, n_] :=
 Module[{m1, m2},
  Sum[Sum[int1[j, m1, m2, m, n] / int2[j, m1, m2],
    {m2, -j, j}], {m1, -j, j}]]
```

For various choices of m, n, and J, the table compares the exact evaluation of $I_J(m, n)$ (equation (14)) with the proposed result $I_C(m, n)$ (equation (15)) and shows that the ratio of the two is a constant, which in this case is $C_0 = \frac{1}{3}$. The evaluation of table[1, 1] takes some time.

```
table[mmax_, nmax_] :=
 Text@
  Grid[
   Prepend[
    Flatten[
     Table[{m, n, j, s[j, m, n], -j (j + 1) ((2 j + 1)^2) a[m, n],
       s[j, m, n] / (-j (j + 1) ((2 j + 1)^2) a[m, n])},
      {m, 0, mmax}, {n, 0, nmax}, {j, 4}], 2],
    {Style["m", Italic], Style["n", Italic],
     Style["J", Italic],
     Row[{Style["I", Italic]Style["J",Italic], "(",
```

```
      Style["m", Italic], ", ", Style["n", Italic], ")"}],
    Row[{Style["I", Italic]Style["C",Italic], "(",
      Style["m", Italic], ", ", Style["n", Italic], ")"}],
    Row[{Style["I", Italic]Style["J",Italic], "(",
       Style["m", Italic], ", ", Style["n", Italic], ")"}] /
     Row[{Style["I", Italic]Style["C",Italic], "(",
       Style["m", Italic], ", ", Style["n", Italic],
       ")"}]}]]
```

```
table[1, 1]
```

m	n	J	$I_J(m,n)$	$I_C(m,n)$	$\frac{I_J(m,n)}{I_C(m,n)}$
0	0	1	-6	-18	$\frac{1}{3}$
0	0	2	-50	-150	$\frac{1}{3}$
0	0	3	-196	-588	$\frac{1}{3}$
0	0	4	-540	-1620	$\frac{1}{3}$
0	1	1	-2	-6	$\frac{1}{3}$
0	1	2	$-\frac{50}{3}$	-50	$\frac{1}{3}$
0	1	3	$-\frac{196}{3}$	-196	$\frac{1}{3}$
0	1	4	-180	-540	$\frac{1}{3}$
1	0	1	$-\frac{3\pi}{2}$	$-\frac{9\pi}{2}$	$\frac{1}{3}$
1	0	2	$-\frac{25\pi}{2}$	$-\frac{75\pi}{2}$	$\frac{1}{3}$
1	0	3	-49π	-147π	$\frac{1}{3}$
1	0	4	-135π	-405π	$\frac{1}{3}$
1	1	1	$-\frac{3\pi}{8}$	$-\frac{9\pi}{8}$	$\frac{1}{3}$
1	1	2	$-\frac{25\pi}{8}$	$-\frac{75\pi}{8}$	$\frac{1}{3}$
1	1	3	$-\frac{49\pi}{4}$	$-\frac{147\pi}{4}$	$\frac{1}{3}$
1	1	4	$-\frac{135\pi}{4}$	$-\frac{405\pi}{4}$	$\frac{1}{3}$

References

[1] D. M. Brink and G. R. Satchler, *Angular Momentum*, Oxford: Clarendon Press, 1979 pp. 21–25.

[2] R. A. Bonham, G. Cooper, and A. P. Hitchcock, “Electron Compton-like Quasielastic Scattering from H_2, D_2, and HD,” *Journal of Chemical Physics*, 130 144303, 2009. dx.doi.org/10.1063/1.3108490.

[3] G. I. Watson, "Neutron Compton Scattering," *Journal of Physics: Condensed Matter*, **8**(33) 5955, 1996. iopscience.iop.org/0953-8984/8/33/005.

R. A. Bonham, “Some Integrals Involving Symmetric-Top Eigenfunctions,” *The Mathematica Journal*, 2012. dx.doi.org/doi:10.3888/tmj.14-12.

About the Author

Russell A. Bonham received his Ph.D. in physical chemistry from Iowa State University in 1957. He is the author of more than 200 articles on all aspects of electron scattering from free atoms and molecules. He is currently a professor emeritus at Indiana University and an adjunct professor at the Illinois Institute of Technology.

R. A. Bonham
Department of Biological, Chemical, and Physical Sciences
Illinois Institute of Technology
3101 South Dearborn Street
Chicago, IL 60616
bonham@iit.edu